

25. Three dimensional plots

Just as in the previous lecture, we start loading the plots package:

```
[> with(plots);
```

25.1 Some basic two and three dimensional plots

Surfaces can be defined as $z = f(x, y)$.

```
[> f := (x,y) -> cos(x*y):
[> plot3d(f,-2*Pi..2*Pi,-2*Pi..2*Pi,axes=box);
```

If we click on the plot, we adjust the orientation and view the object from any angle.

Many surfaces are defined implicitly. For example, the Whitney umbrella is defined by the equation $x^2 - z * y^2 = 0$.

```
[> umbfun := (x,y,z) -> x^2 - z*y^2:
[> umbrella := implicitplot3d(umbfun,-2..2,-2..2,-2..2,grid=[20,10,10]):
[> display(umbrella);
```

The problem with this plot is that the handle of the umbrella is missing. From the equation we see that $x = 0, y = 0$ is a solution for any value of z . But, we do not see the z -axis on the plot.

```
[> handle := plottools[line]([0,0,-2],[0,0,2]):
[> display(whitney,handle);
```

With this patch, we see that the umbrella is really not drawn very well... Maple does not consider the algebraic structure when plotting.

25.2 Options of plot3d

We can arrange the orientation to create a spin plot.

```
[> f := (x,y) -> x^3 - 3*x*y^2:
[> plotargs := f,-1..1,-1..1,axes=box:
[> plot3d(plotargs);
```

The orientation is described by two angles in spherical coordinates.

```
[> plot3d(plotargs,orientation=[45,45]);
```

As you can see, $[45,45]$ is the default orientation. The first angle is the angle with the x -axis, we change this angle to turn the object to the left or right. The second angle is the angle with the z -axis. To see the object from above or below we change this axis.

```
[> spin := [seq(plot3d(plotargs,orientation=[10*i,45]),i=0..36)]:
[> display(spin,insequence=true);
```

As with two dimensional plots, the plotting happens in two stages: first the plot is computed and then rendered. This allows to make lists and sequences of plots, followed by the invocation of the display command.

25.3 Special three dimensional plots

With tubeplot we make a tube around a space curve. Let us make a tubeplot from a circle.

```
[> x := cos(t): y := sin(t): z := 0:
[> torus1 := tubeplot([x,y,z],t=0..2*Pi,radius=1/4):
[> display(torus1);
[> torus2 := tubeplot([x+0.8,z,y],t=0..2*Pi,radius=1/4):
[> display([torus1,torus2]);
```

Let us make a chain of 10 tori :

```
[> chain := []:
[> optstube := t=0..2*Pi,radius=1/4:
[> dx := 0:
[> for i from 1 to 10 do
  >   if i mod 2 = 0
  >     then torus := tubeplot([2*x+dx,2*y,z],optstube):
  >     else torus := tubeplot([2*x+dx,z,2*y],optstube):
  >   end if:
  >   chain := [op(chain),torus]:
  >   dx := dx + 3:
[> end do:
[> display(chain,scaling=constrained);
```

25.4 Data plotting

We can make histograms from matrices:

```
[> indfun := (i,j) -> max(sin(i)*cos(j),0);
[> city := matrix(20,20,indfun):
[> matrixplot(city,heights=histogram);
```

25.5 Animation

Here we will make an animation of a torus knot:

```
[> r := 2 + 4/5*cos(7*t): z := sin(7*t):
[> curve := [r*cos(4*t), r*sin(4*t), z]:
[> picts := [seq(tubeplot(curve,t=0..2*Pi*i/20,radius=1/4),i=1..20)]:
[> display(picts,insequence=true,style=patch);
```

25.6 Assignments

1. Consider the surface defined by $z = \cos(x) \sin(y)$, for $x = -\pi \dots \pi$ and $y = -\pi \dots \pi$.
Give all Maple commands to make a 3D animation of 10 frames that gradually flattens this surface, using the formula

$$\cos((1 - 0.1k)x) \sin(1 - 0.1k)y), \quad \text{for } k = 1, 2, \dots, 10.$$
2. The *Viviani curve* is a space curve defined by $x = R \cos^2(t)$, $y = R \cos(t) \sin(t)$, and $z = R \sin(t)$, where R is some parameter.
 - (a) Give the Maple commands to make a plot of this curve, for $R = 1$, using a tube around the curve of radius 0.05.
 - (b) Make an animation for the curve $x = R \cos^2(t)$, $y = \cos(t) \sin(t)$, and $z = \sin(t)$, for R going from 1 to 10 (thus using ten frames). Give all Maple commands you use to make this animation.